UNIVERSITAIRE FACULTEITEN ST.-IGNATIUS TE ANTWERPEN
SESO
STUDIECENTRUM VOOR ECONOMISCH EN SOCIAAL ONDERZOEK

SIMULATION OF QUEUEING SYSTEMS: A DIDACTICAL NOTE

DE BORGER B. en W. NONNEMAN

Rapport 7983

juni 1979

Universitaire Faculteiten St.-Ignatius
Prinsstraat 13 - 2000 Antwerpen

D/1979/1169/06

This paper is intended as a didactical note on the use of queueing theory and simulation of queueing problems with special reference to ports.

In the first part of this note we review the main results of analytical queueing theory. It will be obvious that the number of analytically tractable systems is fairly limited. Therefore, in part two we discuss simulation routines for some typical queueing systems. Apart from systems description, programs in BASIC for small computers are given in appendix. In part three of this note applications of analytical and simulation procedures are presented.

I. SOME ANALYTICAL MODELS

Probabilistic queueing models are based on the fact that the rate at which customers demand for a service and the rate at which customers can be served by the system is subject to random fluctuation. The structure of 4 typical queueing systems is given in figure 1.

Most well-known analytical models deal with arrivals which are completely random. That is

(1) the probability of a number of arrivals during a certain interval should only depend upon the duration of the interval;

(2) the number of arrivals in a certain interval should be independent of the number of arrivals in any previous interval;

(3) the probability that there is more than one arrival in a small period of time should be very small.

It is shown that if these conditions are satisfied the probability of a number of arrivals during a certain interval follows a Poisson distribution (*). The Poisson distribution is characterized by only one parameter λ, i.e. the average number of arrivals per unit of time. Hence, if this average number of arrivals is known the probability of n arrivals during a time period t is given by

(*) See e.g. Kaufman, A. & Cruon, R., Les phénomènes d'attente, pp.17-22.

Figure 1.


Arrivals
Service station
out of the system
Single channel
Arrivals
Station 1
Station 2
Station k
out of the system
k-stage single channel
Arrivals
Station
Station
Station
out of the system
Multiple parallel channels
Arrivals
Station
Station
Station
out of the system
k-stage multiple parallel channels

$$P(n,t) = \frac{e^{-\lambda t} (\lambda t)^n}{n!}$$

If the probability of the number of arrivals during a certain time period follows a Poisson distribution, it is shown that the relative frequency of time intervals between two arrivals follows an exponential distribution characterized by the inverse of the average number of arrivals. This parameter is interpreted as the average interarrival time.

Consequently, the probability that an elapse of time a between one arrival and the next exceeds a certain value t is given by

$$P(a>t) = e^{-\lambda t}$$

In most analytical models the rate at which ths sytem serves customers is described in an analogous way. If the service rate satisfies the conditions of randomness stated previously, service times are following an exponential distribution defined by one parameter, μ, i.e. the average number of customers served per unit of time. Hence, $p(s>t) = e^{-\mu t}$.

Based on this particular probabilistic description of the behaviour of arriving customers and the technology of service various systems are conceived for which analytical results were derived (*). Three of those systems are discussed below, accompanied by the relevant formulas.

A single exponential channel is the most elementary system. Random arrivals are served as they come with an exponential service time. Arrivals are completely described by the average number of customers per time unit λ and the service station by its average capacity in number of services per time unit μ. The rate of capacity utilization is defined as

---

(*) Only queueing systems with infinite queue capacity and without priority rules are considered. For some other systems, see e.g. Morse, P.M., <u>Queues, inventories and maintenance</u>, pp.116-138.

$\phi = \lambda/\mu$

and average values for interesting descriptive variables can be derived such as waiting time, idle time, queue length, number of units in the system, etc.

Table 1 summarizes some relevant formulas.

Table 1. Single exponential channel: important formulas

- capacity utilization	λ/μ
- expected waiting time	$\frac{\lambda}{\mu(\mu-\lambda)}$
- expected idle time	$(1 - \frac{\lambda}{\mu})\ T$
- expected queue length	$\frac{\lambda^2}{\mu(\mu-\lambda)}$
- expected number of units in the system	$\frac{\lambda}{\mu - \lambda}$
- expected total time in the system	$\frac{1}{\mu} + \frac{\lambda}{\mu(\mu-\lambda)}$
λ = average arrival rate μ = average service rate T = total time period	

The second well-elaborated system consists of a single channel where arrivals are processed through a sequence of services. If in each of the subparts service times follow an exponential distribution with the same average, it can be shown that total service time follows an Erlang k-distribution (*).

(*) The density function of the Erlang k-distribution can be written as: $f(t) = \frac{\lambda^k\ t^{k-1}\ e^{-\lambda t}}{(k-1)!}$

Formulas summarized in table 2, are derived for randomly arriving customers and for an Erlang k-distributed service time.

Table 2. k-stage single channel: important formulas

- capacity utilization	$\phi = \lambda/\mu$
- expected waiting time	$\frac{\phi(k+1)}{2k\mu(1-\phi)}$
- expected idle time	$(1-\phi)T$
- expected queue length	$\frac{\phi^2(k+1)}{2k(1-\phi)}$
- expected number of units in the system	$\frac{2k\phi-\phi^2(k-1)}{2k(1-\phi)}$
- expected total time in the system	$\frac{1}{\mu} + \frac{\phi(k+1)}{2k\mu(1-\phi)}$

λ = average arrival rate
μ = average service rate for the complete system
T = total time

A third class of analytically solved queueing systems are the so-called "parallel channels". In such systems randomly arriving customers are processed by a number of independently operating stations. All stations are identical and have an exponential service time distribution.

Relevant formulas for this system are summarized in Table 3.

Table 3. Parallel channels: important formulas

- capacity utilization	$\phi = \lambda/\mu$
- expected waiting time	$\frac{P_o}{\mu} \cdot \frac{m^m \phi^m}{m!(1-\phi)^2}$
- expected idle time	$(1 - \phi)T$
- expected queue length	$P_o \frac{m^m \phi^{m+1}}{m!(1-\phi)^2}$
- expected number of units in the system	$P_o \frac{m^m \phi^{m+1}}{m!(1-\phi)^2} + m\phi$
- expected total time in the system	$\frac{P_o}{\mu} \cdot \frac{m^m \phi^m}{m!(1-\phi)^2} + \frac{m}{\mu}$
where $P_o^{-1} = \frac{m^m \phi^m}{m!(1-\phi)} + \sum_{n=0}^{m-1} \frac{m^n \phi^n}{n!}$	
λ = average arrival rate	
m = number of channels	
μ = average service rate for the complete system	
T = total time	

II. SIMULATION OF QUEUEING PHENOMENA

It is quite obvious that analytical models are fairly restrictive, considering the required assumptions on the behavior of arrivals and the possibilities of service in order to obtain analytically tractable results.

If applied to port problems the conditions of random arrivals are seldom fulfilled. E.g. seasonal time patterns of commodity flows, liner shipping with fairly regular schedules, etc. lead to dependence of the probability of arrivals during a certain interval upon the exact time and to autoregressive arrival patterns. This often limits the applicability of analytical models to transport shipping.

Similar remarks can be made with regard to the assumptions required for service times.

The idealized systems of analytical models (e.g. equal service rates in parallel stations) are rarely found in practice. E.g. berth capacity often varies with their vintage which leads to non-identical service times and distributions. Furthermore, analytical models are governed by the rule that "any customer" (vessel) can be sent to any service unit ("berth"). It is clear that these assumptions are heroic. At most, analytical models can be used as rough approximations.

Most of these hypothesis can be relaxed by use of simulation processes. In this part we will discuss basic simulation routines of queueing systems. One can easily expand these basic models to suit a particular purpose. Four basic models are presented (*):

1. single station;
2. single proces, consisting of k sequential stations;
3. multiple parallel channels;
4. multiple k-stage parallel channels.

For each of these queueing problems an analytical formulation of the problem with reference to a flow-chart is given. BASIC PROGRAMS for small computers are listed in appendix.

<u>a. Single station (**)</u>

In the formal description of the single station queueing problem (figure 2) we use the following symbols (***):

(*) Slightly other descriptions of some of these models can be found in Naylor e.a., <u>Computer simulation techniques</u>, pp.123-159.

(**) For program see Appendix A.

(***) The same symbols are used in the flow chart. In the programs slight modifications are made e.g. a two letter variable for a vector is not accepted by some BASIC computers.

N = total number of arrivals simulated

A_i = the time interval between the arrival of the (i-1)-th and the i-th customer

$T\emptyset_i$ = the time at which the i-th customer arrives at the system. Consequently $T\emptyset_i = T\emptyset_{i-1} + A_i$

S_i = the time interval required for serving the i-th customer

W_i = the waiting time of the i-th customer

I_i = the idle time of the system while waiting for the i-th customer

$T1_i$ = the time at which the i-th customer leaves the system. Hence, $T1_i = T\emptyset_i + W_i + S_i$.

In block (1) the system is initialized. Waiting time (W), idle time (I) and the time at which the first arrival enters the system (TØ) are set at zero. For the first arrival a service time (S) is generated from an appropriate distribution (*) and the time of departure (T1) is set at this service time.

For the following customers (i=2...N) an interarrival time A_i is generated according to a frequency distribution specified for the particular problems and reflecting the arrival behavior of customers (A_i). The time at which the i-th customer enters the system ($T\emptyset_i$) is calculated in block (3) by adding the interarrival interval (A_i) to the time of entry of the previous customer ($T\emptyset_{i-1}$). In block (4) the time at which the previous customer left the system ($T1_{i-1}$) is compared with the time of entry of the i-th customer. If this departure time of the previous customer exceeds time of arrival of the i-th customer, waiting will be the result as calculated in block (5a). If the reverse is true idle time of the system follows for the difference of departure time and arrival time. This is calculated in block (5c). In block (6) a service time for the i-th

(*) In appendix E BASIC subroutines for the generation of stochastic variates following typical distributions are listed.

Figure 2.


START
INITIALIZE
generate S
T1 = S
w=0
I=0
TØ=0
(1)
(2)
GENERATE A
(3)
TØ = TØ + A
(4)
T1:TØ
T1>TØ
T1<TØ
(5a)
W=T1-TTØ
I=0
T1=TØ
(5b)
W = I = 0
I=TØ-T1
W=0
(5c)
GENERATE S
(6)
T1=TØ+S+W
(7)
CALCULATE
(8)
N=N/1
(9)
STATISTICS

customer (S_i) his generated by an appropriate subroutine so that in block (7) the time of departure for the i-th customer can be calculated viz. the time of departure of the previous customer plus waiting and service time. Block (8) is reserved for calculating a number of interesting statistics such as the sum of waiting times, idle times, time spend in the system, etc. In block (9) the counter for the number of cases simulated is augmented and the procedure is repeated from block (2) on.

The user has to specify the number of arrivals to be simulated or the total-time period the simulation should cover. Output includes some relevant statistics.

b. Sequential process (*)

The previous model is now extended to deal with a k-stage sequential service pattern. Consequently, service times, waiting times, idle times and times of arrival and departure are represented by vectors rather than scalars.

The following definitions are used in flow chart description (figure 3) and program.

N = total number of arrivals simulated

$T\emptyset_{ij}$ = the time at which the i-th customer arrives at the j-th station

A_i = the time interval between the arrival of the (i-1)th customer and the i-th

$T1_{ij}$ = the time at which the i-th customer leaves the system. Hence $T1_{ij} = T\emptyset_{ij} + W_{ij} + S_{ij}$

S_{ij} = the time required to serve the i-th customer at the j-th station

W_{ij} = waiting time of the i-th customer at the j-th station

I_{ij} = idle time of the j-th station waiting for the i-th customer.

(*) For Program see Appendix B.

Figure 3

$$j=1\ldots k \;:\; s_j$$
$$: \; w_j = 0$$
$$T\emptyset_1 = 0 \;:\; T\emptyset_j = \sum_{p=1}^{j-1} Sp$$
$$I_1 = 0 \;:\; I_j = \sum_{p=1}^{j-1} Sp$$
$$T1_j = \sum_{p=1}^{j} Sp$$



In block (1) the system is initialized. For each station j a service time is generated. Waiting times are set to zero. The time at which the first customer arrives at the first station is evidently zero. The first customer arrives at the j-th station after an interval equal to the sum of service times of the (j-1) previous stations. Idle time is zero for the first station and for subsequent stations it is equal to the sum of service times of the (j-1) previous stations.

In block (2) an interarrival time for the next customer is generated. The time at which the next customer enters the system (i.e. the first station) is the previous arrival time (TØ(11)), plus the interarrival time A. This arrival time is set in block (3).

In block (4) to (8) a comparison is made between the time at which the (i-1)th customer left the j-th station and the arrival time of i-th customer in order to determine waiting or idle time. The comparison is made in block (4). If time of departure of the (j-1)th customer surpasses time of arrival of the i-th customer, the i-th customer will have to wait at the j-th station (see 5b). If the reverse holds, the j-th station will be idle for the difference (see 5a). In block (6) a service time for the i-th customer at the j-th station is generated by an appropriate routine. Time of departure from the j-th station is obviously time of arrival plus waiting and service time. This is calculated in block (7). Time of arrival of the i-th customer at the (j+1) station is evidently equal to the time of departure at the j-th station (see block 8).
If for the i-th customer relevant times are calculated at all stations, the next arrival is processed (see block (9)).

If a sufficient number of arrivals is simulated statistics can be derived in block (10).

c. Parallel process (*)

In this section a system with k-parallel stations is described. The basic simulation routine presented here, allocates arrivals to the first available station.

However, with some minor modifications the basic routine can be extended for dealing with priority rules.

We define the following variables:

- N = number of simulated arrivals
- A_i = time elapsed between (i-1)th arrival and i-th arrival
- $TØ_i$ = time at which the i-th customer arrives at the system
- $T1_{ij}$ = time at which the i-th customer leaves the j-th channel
- S_{ij} = the service time for the i-th customer at the j-th channel
- W_{ij} = waiting time for the i-th customer before entering the j-th channel
- I_{ij} = idle time of the j-th channel, waiting for the i-th customer.

In block (1) all times are set to zero. An interarrival time is generated by an appropriate routine in block (2). The time at which the i-th arrival enters the system is determined in block (3). The first available channel is searched in block (4). Search is based on the smallest value of time of departure of the previous customer served by the j-th station. The first available station is denoted by m and consequently the i-th customer will be served by this channel. In block (5) arrival time at the station m is compared with the time of departure of the previous customer served by m. If the time of departure exceeds time of arrival the i-th customer will have to wait (see 6a); if not, the m-th channel will be idle (see 6b). In block (7a) a service time for the m-th channel is generated according to a specified distribution which enables to determine the time of departure of the customer (see block (8)). The procedure is repeated till a sufficient number of arrivals is processed.

(*) For program see Appendix C. For flow-chart description, see fig.4.

Figure 4.


TØ=0
T1(j)=0
W(j)=0
I(j)=0
(1)
INITIALIZE
(2)
GENERATE ARRIVAL A
(3)
TØ = TØ + A
(4)
MIN T1(j) → m
(3)
TØ:T1(m)
TØ<T1(m)
TØ>T1(m)
(6a)
W(m)=TØ-T1(m)
I(m)=W(m)=0
I(m)=T1(m)-TØ
(6b
(7)
GENERATE S
m
(8)
T1(m):TØ+W(m)+S(m)
N=N+1
STATISTICS

d. Sequential-multichannel processes (*)

In practice sequential-multichannel processes are a frequent type of queueing problems. In such a system several channels are available to customers and each customer wants a sequence of services.

The notation in the flow-chart description (fig.5) is similar to the one previously used viz.

N = number of simulated arrivals

A_i = time elapsed between the (i-1)th arrival and the i-th arrival

$S_i(n,k)$ = service time needed for the i-th customer at the k-stage of the n-th channel

$TØ_i(n,k)$ = time of arrival of the i-th customer at the k-th stage of the n-th channel

$T1_i(n,k)$ = time of departure of the i-th customer at the k-th stage of the n-th channel

$I_i(n,k)$ = idle time of the k-th stage facilities of the n-th channel before customer i is served

$W_i(n,k)$ = waiting time of the i-th customer at the k-th stage of the n-th channel.

In block (1) all times are set to zero so that the system is initialized. In block (2) an interarrival time is generated, following a specified distribution. The time at which the i-th customer presents himself to the system i.e. potentially the first stage of each channel equals the time of arrival of the previous customer plus the interarrival time. This is calculated in block (3). In block (4) the first available channel is determined. Times of departure of a previous customer from the first stage of each channel are checked and the customer is designated to the m-th channel, i.e. the first available one.

(*) For program see Appendix D.

Figure 5.


(1)
INITIALIZE
(2)
GENERATE A
(3)
for each n
TØ(n,1)=TØ(n,1)+A
(4)
min
T1(n,1)
→ m
for each k
(5)
TØ>T1
TØ(m,k):T1(m,k)
TØ<T1
(6a)
I(m,k)=TØ(m,k)-T1(m,k)
I(m,k)=W(m,k)=0
W(m,k)=T1(m,k)-TØ(m,k)
(6b)
(7)
S(m,k)
(8)
T1(m,k)=TØ(m,k)+W(m,k)+S(m,k)
TØ(m,k+1)=T1(m,k)
N=N+1
Statistics

In block (5) to (8) relevant times are calculated for each stage in the m-th channel. In (5) arrival time at the stage of the i-th customer is compared with departure time of the previously served customer and either idle time or waiting time is determined in blocks (6). Next, a service time for the relevant stage is generated in block (7), after which time of departure in the k-th stage and time of arrival in the (k+1)-th stage can be found (see 8). If the customer is processed through all stages of the m-th channel the procedure is repeated for another arrival.

It is clear that the sequential-parallel process is a generalization of the cases discussed in (a),(b) and (c). The program developed for this general process can be used - although it will require a slightly higher processing time - for these simpler queueing systems.

III. APPLICATIONS IN PORT ECONOMICS

In this part examples of both analytical and simulation methods are discussed (*). In most practical problems analytical methods can be used as rough-and-ready approximations. However, from the applications it will be clear that the possibilities of simulation methods are enormous. With more creativity the user can fit the basic routines discussed in the previous sections to extremely complicated real world situations. By simulation both an understanding of the real world and an quantitive insight in the relations between inputs and outputs can be acquired.

(*) The examples are only given as an illustration of simulation procedures to queueing phenomena. This doesn't mean, however, that the applicability of simulation techniques is restricted to the type of problems discussed. For a survey of applications in ports see Robinson, R. & Tognetti, K.P., Modelling and Port Policy Decisions: the Interface of Simulation and Practice, Proceedings of the First International Congress in Transportation Research, Bruges, 1973.

a. An application of the single station procedure

Consider the following situation. An inland port is accessible through an entrance channel. Along this channel a small section is fairly narrow so that for safety reasons only one vessel at the time is allowed to use that section (see fig.5).

Figure 5.



One could consider this section as a "station". The service time would be the time required by a vessel to pass the section. Service time depends upon speed, which is related to vessel size. Vessel sizes are often exponentially distributed. Consequently one may safely assume that "service times" are also exponentially distributed. Average time of passage is estimated at 30 minutes per vessel. Using analytical formulas to tackle this problem implies that arrivals are random.

Let us first assume that this is the case and estimate expected congestion by using analytical formulas.

Evidently the single exponential channel formulas are appropriate here. Hence, average waiting time per vessel (W)

$$W = \frac{\lambda}{\mu(\mu-\lambda)}$$

where λ is the average arrival rate and μ is the average service rate. With 30 minutes per vessel as average service time and knowing that the approach is available 24 hours a day average service rate is about 48 vessels per day. Waiting time in hours in function of the average traffic per day (x) is given by

$$W = \frac{x}{96 - 2x}$$

This formula is plotted in Figure 6.

It is clear that as traffic approaches technical capacity average waiting times increase at a very fast rate.

On the same figure simulated average waiting times are indicated by triangles. The program of Appendix A was used and 1000 arrivals were simulated.

Table 5. summarizes the main results of the simulation.

Table 5.

	Average arrival rate vessels/ day	Average inter-arrival time	Utilization rate in percent	Average waiting time	Period simulated
Case I	10	2h24min	21.6	8min	101 days
Case II	20	1h12min	29.9	17min	51 days
Case III	30	48min	60.4	43min	34 days
Case IV	40	36min	83.2	3h53min	25 days

These results can be compared with the theoretical values. It is clear that if traffic approaches capacity the simulation on the basis of a fixed number of arrivals becomes rather imprecise. This is partly due to the fact that the period over which the simulation extends becomes rather short. For high utilization rates it is recommended that more arrivals are simulated to extend the total time period of the simulation excercise.

Figure 6.


Average waiting time (hours)
4h
3h30min
3h
2h30min
2h
1h30min
1h
30min
0
10
20
30
40
50
Average number of vessels per day
Capacity (vessels per day)

Information on congestion is extremely useful, e.g. for the evaluation of capacity extentions. Suppose a dredging and enlargement project is considered by the port authorities, a rough approximation of benefits is found by using queueing methods.

Say that the program would cost I and double existing capacity to a maximum of 96 vessels a day.

The effect on average waiting time is easily derived, as average waiting time per vessel (in hours) now goes from

$$\frac{x}{96-2x} \quad \text{to} \quad \frac{x}{384-4x}$$

Daily savings from this investment are consequently the difference between total waiting costs in both cases. This equals

$$k \left[\frac{x^2}{96 - 2x} - \frac{x^2}{384 - 4x} \right]$$

where k = average cost per hour of waiting of a vessel and
x = the average number of vessels a day.

If anually say X thousand vessels are expected, after some algebra one finds that annual savings are

$$2740\ k\ X^2 \left[\frac{1}{96 - 5.48\ X} - \frac{1}{384 - 10.96\ X} \right]$$

The net present value (NPV) of such an investment (assuming constant traffic over time and over a very long period) would be

$$NPV = \frac{2740\ k\ X^2}{r} \left[\frac{1}{96 - 5.48X} - \frac{1}{384 - 10.96X} \right] - I$$

where r is an appropriate discount rate.

If the NPV is positive an investment of the amount I is justified. In Table 6 the maximum acceptable amount of investment is given for several annual levels of traffic (10 % discount rate and k=1000 per hour).

Table 6.

Number of vessels per year	Permissable investment million $
7.500	23
10.000	57
12.500	138

Investment is not necessarily the cheapest way to reduce congestion and its cost. Simple organizational procedures are often more effective to curb congestion. Simulation methods can also be used to evaluate the impact of organizational schedules on congestion.

Suppose that in the foregoing problem of the entrance canal, departure can to some extent be scheduled. Each in-port vessel is given a target hour of departure (i.e. a target time of arrival at the bottle neck of the entrance channel). The target hour should be respected within an interval of plus or minus 1 hour. Consequently, when a departing vessel is considered, interarrival time in hours will be situated between

$$\frac{24}{x} - 1 \quad \text{and} \quad \frac{24}{x} + 1$$

with x the daily number of vessels traversing the entrance channel. We assume that out-of-port-vessels arrive at random, so that their interarrival rates are exponentially distributed with an average time of $\frac{24}{x}$.

How to handle such a situation with the single station simulation model? The only problem is to construct an interarrival time generator which reflects the behavior of arrivals. In Figure 7 a flow chart for such a generator is represented. In block (1) a random number, uniformly distributed between 0 and 1 is generated. Half of the time its value will be below 1/2 and half of the time its value is above 1/2. If the random number is below 1/2, the vessel arriving at the bottleneck is inbound; a random number exceeding 1/2 means that the arriving vessel goes outbound. This is determined in block (2). If the arrival is inbound an exponentially distributed interarrival time is generated (3b). If it concerns an outbound arrival a normally distributed interarrival time is generated (3a). The appropriate value is returned to program.

Figure 7.



(*) This interarrival time generator will generate about 95 % of the outbound arrivals within the allowed interval of 2 hours, viz. between -2σ and 2σ.

In Figure 8 the results of the simulation excercise are plotted. In this figure the simulation outcomes with the departure schedule are marked by a circle; the simulation outcomes without the departure schedule are denoted by a triangle.

One can see the impact of this organizational method of reducing congestion. The gain is small with moderate levels of demand. However as soon as the number of arrivals gets closer to capacity, important gains are realised. The importance of the gain can be illustrated by estimating the equivalent investment in capacity necessary to bring down congestion to the same level.

One remembers that waiting time is given by

$$W = \frac{\lambda}{\mu(\mu-\lambda)}$$

with λ = arrival rate and
μ = service rate.

Hence waiting time in terms of traffic and capacity is

$$W = \frac{24\ x}{C(C-x)}$$

with x = traffic level per day
C = capacity per day and
W = waiting time in hours.

If traffic is at a level of 40 passages a day, one knows that with the departure schedule average waiting time is about 1 hour and 37 minutes. Without a departure schedule this would require a capacity of

$$C = 1/2\ x + 1/2\ \sqrt{x^2 + 96\ x/W} = 51.5$$

Figure 8.


Average waiting time (hours)
4h
3h30min
3h
2h30min
2h
1h30min
1h
0
10
20
30
40
50
Average number of vessels per day
Capacity (vessels per day)

which means that the equivalent capacity extention would be about 7.5 percent (*).

By considering the trade-off between different organizational schedules and the equivalent capacity investment a fairly good idea of the monetary value of good managerial procedures can be obtained.

b. An application of the multiple channel procedure

Suppose two terminals are available for handling a particular type of traffic. An investment in an additional terminal is considered because of increased demand. Two options are possible:

1. build a terminal with the same capacity;
2. build a modernly equipped facility with a capacity of about double the capacity of the existing terminals. Apart from other costs and benefits for each option, an essential input in an investment analysis is the effect on congestion in each of the proposals.

The first option - i.e. terminals of equal capacity - can be handled by analytical formulas provided that arrival and service rates are randomly fluctuating with an appropriate average. The existing situation can also be analysed by these formulas. However, the second option with differential service rates can only be tackled through a simulation approach.

Suppose existing terminals each have an annual capacity of about 750.000 tons. Average waiting time for the actual situation is derived by using the multiple parallel channel formulas of section 1. Calculated and simulated values for average waiting time per vessel (assume an everage vessel size of 2500 tons) are summarized in Table 6.

(*) Of course, it's possible to analyse the sensitivity of waiting time (and the corresponding value of C) with regard to different organizational schedules (e.g. one could simulate the impact on congestion when narrowing the arrival interval).

Table 6. Existing situation

Traffic (million of tons)	Utilisation rate	Average waiting time per vessel	
		Calculated	Simulated (*)
0,5	33 %	3h39min	3h45min
0,75	50 %	9h51min	9h58min
1	67 %	23h22min	23h51min
1,25	89 %	2days18h26min	3days4h39min

(*) Simulation results based on a total period of 3000 units of time. Of course, if more precision is required (especially for high utilisation rates), one can improve the accuracy of the results by simulating a longer time period.

A similar procedure is followed to estimate the impact on congestion of an additional terminal of 3/4 million tons capacity. Calculated and simulated waiting times are presented in Table 7 for various levels of traffic.

Table 7. Option 1

Traffic (million of tons)	Utilisation rate	Average waiting time per vessel	
		Calculated	Simulated
0,5	22 %	26min	27min
0,75	33 %	1h19min	1h28min
1	44 %	3h04min	2h51min
1,25	56 %	6h34min	6h27min
1,5	67 %	12h55min	11h34min
1,75	88 %	1day2h43min	1day4h16min
2	89 %	2days21h52min	2days16h5min

Differential service times can only be tackled by a simulation procedure. Table 8 summarizes the results of a simulation run on a fixed period of 3000 units of time.

Table 8. Option 2

Traffic (million of tons)	Utilisation rates		Average waiting times (simulated)
	Stations 1,2	Station 3	
0,5	23 %	13 %	11min
0,75	30 %	20 %	23min
1	37 %	28 %	1h06min
1,25	50 %	35 %	1h50min
1,5	55 %	45 %	2h56min
1,75	66 %	54 %	5h50min
2	70 %	60 %	11h40min
2,25	79 %	71 %	18h15min
2,5	86 %	80 %	1day3h01min
2,75	92 %	89 %	2days10h46min

Congestion impacts of the different investment schemes are compared with the actual situation in Figure 9. As before simulated results are indicated by triangles.

Careful simulation provides valuable information to estimate the monetary gains of reduced congestion by expansion. This information is essential for evaluating investment proposals.

Figure 9.


Average waiting times
70h
60h
50h
40h
30h
20h
10h
Traffic per year (million tons)
0,5
0,75
1
1,25
1,5
1,75
2
2,25
2,5
2,75
3

IV. SUMMARY

In the first section of this note the major analytical results for queueing problems and the accompanying assumptions were presented. In section II simulation methods for single, sequential and multiple stations problems are discussed with BASIC programs (see Appendix). Section III explores some possible areas of application in ports.

REFERENCES

KAUFMAN, A. & CRUON, E., Les phénomènes d'attente, Dunod, Paris, 1962.

METTAM ; J.D., Forecasting Delays to Ships in Port, Dock & Harbour Authority, April 1967.

MORESE, P.M., Queues, inventories and maintenance, John Wiley, & Sons, New York, 1958.

NAYLOR, T.A., BALUITFY, J.L., BURDICK, D.S. & CHU, K., Computer Simulation Techniques, John Wiley & Sons, New York, 1966.

ROBINSON, R. & TOGNETTI, K.P., Modelling and Port Policy Decisions: the interface of simulation and practice, Proceedings of the first International Congress in Transportation Research, Bruges, 1973.

WHITE, I.R.P., The use of waiting time theory in planning expansion of port facilities, Dock & Harbour Authority, Febr.1972.

APPENDIX A.

```
FILE     W1

0010 REM *** SINGLE STATION ***
0020 REM PARAMETERS OF ARRIVAL AND SERVICE TIME DISTRIBUTIONS
0030 INPUT A0,S0
0040 REM INPUT NUMBER OF SIMULATIONS
0050 INPUT N9
0060 REM INTIALIZE
0070 LET P1=0
0080 LET P2=0
0090 LET N=0
0100 LET U=0
0110 LET W=0
0120 LET I=0
0130 GOSUB 370
0140 LET V=S
0150 REM START SIMULATING
0160 LET N=N+1
0165 RANDOMIZE
0170 GOSUB 420
0180 LET U=U+A
0190 REM CHECK FOR WAITING OR IDLE TIME
0200 IF V>U THEN 240
0210 LET I=U-V
0220 LET W=0
0230 GOTO 260
0240 LET W=V-U
0250 LET I=0
0260 REM GENERATE SERVICE TIME AND CALCULATE PARAMETERS
0270 GOSUB 370
0280 LET V=U+S+W
0290 LET P1=P1+W
0300 LET P2=P2+I
0310 IF N<N9 THEN 160
0320 REM PRINT STATISTICS
0330 PRINT "AVERAGE WAITING TIME",P1/N
0340 PRINT "UTILISATION RATE",1-P2/V
0350 PRINT "TOTAL TIME SIMULATED",V
0360 GOTO 470
0370 REM SERVICE TIME GENERATOR
0390 LET R=RND
0400 LET S=-S0*LOG(R)
0410 RETURN
0420 REM ARRIVAL TIME GENERATOR
0430 LET R=RND
0450 LET A=-A0*LOG(R)
0460 RETURN
0470 END

END OF LISTING
```

APPENDIX B.

FILE   W2

```
0010 REM *** SEQUENTIAL PROCESS ***
0020 DIM U[11],V[11],W[10],I[10],S[10]
0030 REM N9=NR. OF SIMULATIONS; K=NR. OF STAGES
0040 REM A0=PARAMETER OF ARRIVAL DISTRIBUTION
0050 REM S[J]=PARAMETERS OF SERVICE DISTRIBUTIONS
0060 DIM U[11],V[11],W[10],I[10],S[10]
0070 DATA 1000,3
0080 DATA 5
0090 DATA 1,1,1
0100 READ N9,K
0110 READ A0
0120 FOR J=1 TO K STEP 1
0130 READ S[J]
0140 NEXT J
0150 REM INTITIALIZE
0160 LET Z=0
0170 LET N=0
0180 LET U[1]=0
0190 LET I[1]=0
0200 FOR J=1 TO K STEP 1
0210 LET W[J]=0
0220 GOSUB 660
0230 LET Z=Z+S9
0240 LET U[J+1]=Z
0250 LET I[J+1]=Z
0260 LET V[J]=Z
0270 NEXT J
0280 REM START SIMULATING
0290 LET N=N+1
0300 RANDOMIZE
0310 GOSUB 620
0320 LET U[1]=U[1]+A
0330 REM CHECK EACH STAGE FOR WAITING OR IDLE TIMES
0340 FOR J=1 TO K STEP 1
0350 IF U[J]>V[J] THEN 390
0360 LET W9=V[J]-U[J]
0370 LET I9=0
0380 GOTO 410
0390 LET I9=U[J]-V[J]
0400 LET W9=0
0410 LET I[J]=I[J]+I9
0420 LET W[J]=W[J]+W9
0430 REM GENERATE SERVICE TIME
0440 GOSUB 660
0450 REM CALCULATE DEPARTURE TIMES
0460 LET V[J]=U[J]+W9+S9
0470 LET U[J+1]=V[J]
0480 NEXT J
0490 IF N<N9 THEN 290
0500 REM PRINT STATISTICS
```

```
0510 PRINT "AVERAGE WAITING TIMES"
0520 FOR J=1 TO K STEP 1
0530 PRINT J,W[J]/N
0540 NEXT J
0550 PRINT
0560 PRINT "UTILISATION RATES"
0570 FOR J=1 TO K STEP 1
0580 PRINT 1-I[J]/V[J]
0590 NEXT J
0600 PRINT "TOTAL TIME SIMULATED",V[K]
0610 GOTO 700
0620 REM ARRIVAL TIME GENERATOR
0630 LET R=RND
0640 LET A=-A0*LOG[R]
0650 RETURN
0660 REM SERVICE TIME GENERATOR
0670 LET R=RND
0680 LET S9=-S[J]*LOG[R]
0690 RETURN
0700 END

END OF LISTING
```

APPENDIX C.

FILE W3

```
0010 REM *** MULTIPLE CHANNEL ***
0020 DIM V[10],W[10],I[10],S[10],F[10]
0030 REM N9=NR. OF SIMULATIONS, K=NR. OF CHANNELS
0040 REM A0=PARAMETER OF ARRIVAL DISTRIBUTION
0050 REM S[J]=PARAMETERS OF SERVICE DISTRIBUTION
0060 DATA 100,3
0070 DATA 5
0080 DATA 5,5,5
0090 READ N9,K
0100 READ A0
0110 FOR J=1 TO K STEP 1
0120 READ S[J]
0130 NEXT J
0140 REM INITIALIZE
0150 LET U=0
0160 LET N=0
0170 FOR J=1 TO K STEP 1
0180 LET V[J]=0
0190 LET W[J]=0
0200 LET I[J]=0
0210 LET F[J]=0
0220 NEXT J
0230 REM START SIMULATING
0240 LET N=N+1
0250 RANDOMIZE
0260 GOSUB 620
0270 LET U=U+A
0280 REM CHECK FOR FIRST AVAILABLE STATION
0290 LET M=1
0300 LET Z=V[1]
0310 FOR J=2 TO K STEP 1
0320 IF V[J]>Z THEN 350
0330 LET Z=V[J]
0340 LET M=J
0350 NEXT J
0360 REM COUNT NUMBER OF CUSTOMERS PROCESSED
0370 LET F[M]=F[M]+1
0380 REM CHECK FOR WAITING OR IDLE TIMES
0390 IF U>V[M] THEN 430
0400 LET W9=V[M]-U
0410 LET I9=0
0420 GOTO 450
0430 LET I9=U-V[M]
0440 LET W9=0
0450 GOSUB 660
```

```
0460 REM CALCULATE STATISTICS AND DEPARTURE TIME
0470 LET V[M]=U+W9+S9
0480 LET W[M]=W[M]+W9
0490 LET I[M]=I[M]+I9
0500 IF N<N9 THEN 240
0510 REM PRINT STATISTICS
0520 PRINT "AVERAGE WAITING TIMES"
0530 FOR J=1 TO K STEP 1
0540 PRINT J,W[J]/F[J]
0550 NEXT J
0560 PRINT
0570 PRINT "UTILISATION RATES"
0580 FOR J=1 TO K STEP 1
0590 PRINT J,1-I[J]/V[J]
0600 NEXT J
0610 GOTO 700
0620 REM ARRIVAL TIME GENERATOR
0630 LET R=RND
0640 LET A=-A0*LOG(R)
0650 RETURN
0660 REM SERVICE TIME GENERATOR
0670 LET R=RND
0680 LET S9=-S[M]*LOG(R)
0690 RETURN
0700 END

END OF LISTING
```

APPENDIX D.

FILE W4

```
0010 REM *** SEQUENTIAL-MULIPLE PROCESS ***
0020 DIM U[5,5],V[5,5],W[5,5],I[5,5],S[5,5],F[5]
0030 REM N9=NR. OF SIMULATIONS, K1=NR. OF CHANNELS, K2=NR. OF STAGES
0040 REM A0=PARAMETER OF ARRIVAL DISTRIBUTION
0050 REM S[I,J]=PARAMETERS OF SERVICE DISTRIBUTION
0060 DATA 100,2,2
0070 DATA 5
0080 DATA 5,2.5
0090 DATA 5,2.5
0100 READ N9,K1,K2
0110 READ A0
0120 FOR J1=1 TO K1 STEP 1
0130 FOR J2=1 TO K2 STEP 1
0140 READ S[J1,J2]
0150 NEXT J2
0160 NEXT J1
0170 REM INITIALIZE
0180 LET N=0
0190 FOR J1=1 TO K1 STEP 1
0200 LET F[J1]=0
0210 FOR J2=1 TO K2 STEP 1
0220 LET U[J1,J2]=0
0230 LET V[J1,J2]=0
0240 LET W[J1,J2]=0
0250 LET I[J1,J2]=0
0260 NEXT J2
0270 NEXT J1
0280 REM START SIMULATING
0290 LET N=N+1
0300 GOSUB 780
0310 REM CALCULATE ARRIVAL TIME AT EACH CHANNEL
0320 FOR J1=1 TO K1 STEP 1
0330 LET U[J1,1]=U[J1,1]+A
0340 NEXT J1
0350 REM SEARCH FIRST AVAILABLE STATION
0360 LET M=1
0370 LET Z=V[1,1]
0380 FOR J1=1 TO K1 STEP 1
0390 IF V[J1,1]>Z THEN 420
0400 LET Z=V[J1,1]
0410 LET M=J1
0420 NEXT J1
0430 REM COUNT NUMBER OF CUSTOMES PROCESSED
0440 LET F[M]=F[M]+1
```

```
0450 REM CHECK FOR WAITING OR IDLE TIMES
0460 FOR J2=1 TO K2 STEP 1
0470 IF U[M,J2]>V[M,J2] THEN 510
0480 LET W9=V[M,J2]-U[M,J2]
0490 LET I9=0
0500 GOTO 530
0510 LET W9=0
0520 LET I9=U[M,J2]-V[M,J2]
0530 GOSUB 820
0540 REM CALCULATE STATISTICS AND DEPARTURE TIME
0550 LET W[M,J2]=W[M,J2]+W9
0560 LET I[M,J2]=I[M,J2]+I9
0570 LET V[M,J2]=U[M,J2]+W9+S9
0580 LET U[M,J2+1]=V[M,J2]
0590 NEXT J2
0600 IF N<N9 THEN 290
0610 PRINT "AVERAGE WAITING TIMES"
0620 FOR J1=1 TO K1 STEP 1
0630 PRINT J1
0640 FOR J2=1 TO K2 STEP 1
0650 PRINT J1;J2;W[J1,J2]/F[J1],
0660 NEXT J2
0670 PRINT
0680 NEXT J1
0690 REM PRINT STATISTICS
0700 PRINT "UTILISATION RATES"
0710 FOR J1=1 TO K1 STEP 1
0720 FOR J2=1 TO K2 STEP 1
0730 PRINT J1;J2;1-I[J1,J2]/V[J1,J2],
0740 NEXT J2
0750 PRINT
0760 NEXT J1
0770 GOTO 860
0780 REM ARRIVAL TIME GENERATOR
0790 LET R=RND
0800 LET A=-A0*LOG[R]
0810 RETURN
0820 REM SERVICE TIME GENERATOR
0830 LET R=RND
0840 LET S9=-S[M,J2]*LOG[R]
0850 RETURN
0860 END

END OF LISTING
```

APPENDIX E.

FILE

```
0010 REM NEGATIVE EXPONENTIAL DISTRIBUTION
0020 LET R=RND
0025 REM MEAN OF ARR. OR SERV.DISTRIBUTION:E[J]
0030 LET S=-E[J]*LOG[R]
0040 RETURN
```

END OF LISTING

FILE

```
0010 REM ERLANG-K DISTRIBUTION
0020 LET P=1
0030 FOR B=1 TO K STEP 1
0040 LET R=RND
0050 LET P=P*R
0060 NEXT B
0070 LET S=-LOG[P]/[K/E[J]]
0080 RETURN
```

END OF LISTING

FILE

```
0010 REM NORMAL DISTRIBUTION
0020 LET P=0
0030 FOR B=1 TO 12 STEP 1
0040 LET R=RND
0050 LET P=P+R
0060 NEXT B
0065 REM STANDARD DEVIATION:S[X]
0070 LET S=S[X]*[P-6]+E[J]
0080 RETURN
```

END OF LISTING